

Esercizi di applicazione

Lorenzo Roi

Copyright 2006 www.lorenzoroi.net

■ Istruzioni

Si presentano diversi esercizi dai più semplici a quelli più impegnativi utilizzando le funzioni del pacchetto **RettaFasci.m**.

Cliccare qui per caricare il pacchetto

Modificare il percorso sottostante sostituendolo con quello effettivo contenente il pacchetto **RettaFasci.m** (ponendo attenzione alle maiuscole-minuscole e alla lettera dell'unità) e quindi avviare il calcolo delle tre celle successive cliccando sul pulsante sopra.

```
| SetDirectory["C:/Documents and Settings/Lorenzo/Documenti/rette"];  
  
| Needs["Miscellaneous`RealOnly`"]  
| << RettaFasci.m
```

I successivi due comandi eliminano gli avvisi riguardante la definizione e l'uso di termini con uno spelling simile.

```
| Off[General::spell];  
| Off[General::spell1];
```

Dopo ogni esercizio, prima di passare al successivo, conviene cancellare i simboli usati attivando la cella posta prima del testo.

1. Esercizi elementari

Esercizio 1.1

```
| Clear["Global`*"]
```

▼ Determinare la retta s per $A(1, 2)$ parallela alla retta $r : 2x - 3y - 4 = 0$.

Definiti gli elementi dati dal problema

```
| puntoA = {1, 2};  
| r = 2 x - 3 y - 4 == 0;
```

possiamo sfruttare direttamente la funzione

```
| s1 = rettaParallela[puntoA, r]
```

$$4 + 2x - 3y = 0$$

oppure, individuare il coefficiente angolare di r

```
| mr = coefficienteAngolare[r]
```

$$\frac{2}{3}$$

e quindi procedere con

```
| s2 = rettaParallela[puntoA, mr]
```

$$-\frac{4}{3} - \frac{2x}{3} + y = 0$$

Le due rette coincidono: difatti riportate entrambe alla forma esplicita

```
| equazioneEsplicita /@ {s1, s2}
```

$$\left\{ y = \frac{4}{3} + \frac{2x}{3}, y = \frac{4}{3} + \frac{2x}{3} \right\}$$

otteniamo la medesima equazione.

Esercizio 1.2

```
| Clear["Global`*"]
```

▼ Determinare la retta t per $A(1, 2)$ perpendicolare al segmento di estremi $B(-1, -2)$, $C(2, -3)$.

Definiti gli elementi del problema

```
| {puntoA = {1, 2},  
  puntoB = {-1, -2},  
  puntoC = {2, -3}}
```

$$\{\{1, 2\}, \{-1, -2\}, \{2, -3\}\}$$

il coefficiente angolare della retta BC risulta

```
| mBC = coefficienteAngolare[puntoB, puntoC]
```

$$-\frac{1}{3}$$

cosicché la retta richiesta è

```
| t = rettaPerpendicolare[puntoA, mBC]
```

$$1 - 3x + y = 0$$

Nella forma esplicita

```
| equazioneEsplicita[t]
```

$$y = -1 + 3x$$

appare evidente che il suo coefficiente angolare è l'opposto del reciproco di m_{BC} .

Esercizio 1.3

```
| Clear["Global`*"]
```

▼ Determinare il punto simmetrico di $A(-2, 3)$ rispetto alla retta $r : 4x - 2y + 1 = 0$.

Definiti punto A e retta r

```
| {puntoA = {-2, 3},  
  r = 4 x - 2 y + 1 == 0}
```

$$\{-2, 3\}, 1 + 4x - 2y = 0$$

determiniamo la retta s perpendicolare a r per A

```
| s = rettaPerpendicolare[puntoA, r]
```

$$8 - 2x - 4y = 0$$

Questa retta interseca r in M

```
| puntoM = puntoIntersezione[r, s]
```

$$\left\{ \frac{3}{5}, \frac{17}{10} \right\}$$

cosicché, detto $P_0(x_0, y_0)$ il punto simmetrico di A

```
| puntoP0 = {x0, y0}
```

$$\{x_0, y_0\}$$

le sue coordinate devono soddisfare alla condizione

```
| eq = puntoM == puntoMedio[puntoA, puntoP0]
```

$$\left\{ \frac{3}{5}, \frac{17}{10} \right\} = \left\{ \frac{1}{2} (-2 + x_0), \frac{3 + y_0}{2} \right\}$$

Questo risultato costituisce un sistema di due equazioni: risolto fornisce il punto cercato.

```
| sol = Solve[eq, {x0, y0}]
```

$$\left\{ \left\{ x_0 \rightarrow \frac{16}{5}, y_0 \rightarrow \frac{2}{5} \right\} \right\}$$


Abbiamo quindi che

```
| puntoP0 = Flatten[puntoP0 /. sol]
```

$$\left\{ \frac{16}{5}, \frac{2}{5} \right\}$$

La situazione grafica discussa è rappresentata sotto

```
| Show[graficoRette[{r, s}, {x, -5, 5}, {y, -5, 5}, PlotStyle -> colore /@ {rosso, verde},
  DisplayFunction -> Identity], Graphics[{PointSize[0.02], Point /@ {puntoA, puntoM, puntoP0}}],
  Axes -> True, AspectRatio -> Automatic, DisplayFunction -> $DisplayFunction];
```



2. Esercizi di primo livello

Esercizio 2.1

```
| Clear["Global`*"]
```

- ▼ Verificare che i punti $A(1, 2)$, $B(-2, 1)$ e $C(0, -2)$ non sono allineati e che quindi determinano un triangolo, di cui si chiedono:
- le equazioni dei lati;
 - le equazioni delle parallele ai lati condotte dai vertici opposti;
 - il perimetro e l'area;
 - il circocentro.

Definiti i tre punti

```
{puntoA = {1, 2},
 puntoB = {-2, 1},
 puntoC = {0, -2}}
```

```
{{1, 2}, {-2, 1}, {0, -2}}
```

analizziamo se essi sono collineari ossia

```
puntiCollineari[puntoA, puntoB, puntoC]
```

```
False
```

a) Non essendo allineati, determiniamo le equazioni delle rette comprendenti i lati

```
{rettaAB = retta2Punti[puntoA, puntoB],
 rettaBC = retta2Punti[puntoB, puntoC], rettaCA = retta2Punti[puntoC, puntoA]}
```

```
{5 + x - 3 y == 0, 4 + 3 x + 2 y == 0, 2 - 4 x + y == 0}
```

b) Le rette parallele a queste per il vertice opposto si ottengono invece come

```
{parallelaAB = rettaParallela[puntoC, rettaAB],
 parallelaBC = rettaParallela[puntoA, rettaBC], parallelaCA = rettaParallela[puntoB, rettaCA]}
```

```
{-6 + x - 3 y == 0, -7 + 3 x + 2 y == 0, -9 - 4 x + y == 0}
```

c) le lunghezze dei lati risultano

```
{latoAB = distanza2Punti[puntoA, puntoB],
 latoBC = distanza2Punti[puntoB, puntoC], latoCA = distanza2Punti[puntoC, puntoA]}
```

```
{ $\sqrt{10}$ ,  $\sqrt{13}$ ,  $\sqrt{17}$ }
```

per cui il perimetro vale

```
perimetro = latoAB + latoBC + latoCA
```

```
 $\sqrt{10} + \sqrt{13} + \sqrt{17}$ 
```

oppure, in forma alternativa

```
| Apply[Plus, lati]
```

$$\sqrt{10} + \sqrt{13} + \sqrt{17}$$

L'area si può determinare con la formula che la esprime in funzione delle coordinate dei vertici

```
| areaABC = areaTriangolo[puntoA, puntoB, puntoC]
```

$$\frac{11}{2}$$

oppure trovando la distanza di un vertice (altezza) dal lato opposto (base)

```
| altezzaCH = distanzaPuntoRetta[puntoC, rettaAB]
```

$$\frac{11}{\sqrt{10}}$$

e quindi

```
| areaABC = 1/2 latoAB altezzaCH
```

$$\frac{11}{2}$$

Infine il circocentro si individua come l'intersezione degli assi di due lati. Determinati i punti medi

```
| {puntoMedioAB = puntoMedio[puntoA, puntoB], puntoMedioBC = puntoMedio[puntoB, puntoC]}
```

$$\left\{ \left\{ -\frac{1}{2}, \frac{3}{2} \right\}, \left\{ -1, -\frac{1}{2} \right\} \right\}$$

i due assi sono

```
| {asseAB = rettaPerpendicolare[puntoMedioAB, rettaAB],  
  asseBC = rettaPerpendicolare[puntoMedioBC, rettaBC]}
```

$$\left\{ -3x - y = 0, \frac{1}{2} + 2x - 3y = 0 \right\}$$

e la loro intersezione fornisce il circocentro

```
| circocentro = puntoIntersezione[asseAB, asseBC]
```

$$\left\{ -\frac{1}{22}, \frac{3}{22} \right\}$$

Esercizio 2.2

```
| Clear["Global`*"]
```


- ▼ Sono date le rette $a: y = 3x + 1$ e $b: y = \frac{1}{3}x + 1$ e siano c e d le bisettrici degli angoli formati da a e b , rispettivamente di coefficiente angolare positivo e negativo. Determinare il rapporto dell'area del triangolo formato da a , d e dall'asse x , con l'area del triangolo definito da b , c e l'asse x .

Dopo averle definite,

$$\begin{cases} a = y = 3x + 1; \\ b = y = \frac{1}{3}x + 1; \end{cases}$$

proponiamo innanzitutto un disegno delle rette a e b

```
| graficoRette[{a, b}, {x, -2, 2}, {y, -1, 3}, PlotStyle -> {colore[rosso], colore[verde]}];
```



Entrambe passano per il punto $R(0, 1)$ e possiedono coefficienti angolari uno reciproco dell'altro (vedere [alla fine dell'esercizio](#) per un metodo alternativo). Essendo le bisettrici i luoghi dei punti equidistanti dalle rette determiniamo le equazioni di queste due rette che, per note proprietà di geometria elementare, risulteranno mutuamente perpendicolari. Definito quindi un punto generico $P(x, y)$

```
| puntoP = {x, y};
```

l'equazione che ne risulta ponendo uguali le sue distanze dalle due rette è

```
| equaz = distanzaPuntoRetta[puntoP, a] == distanzaPuntoRetta[puntoP, b]
```

$$\frac{\text{Abs}[-1 - 3x + y]}{\sqrt{10}} = \frac{3 \text{Abs}[-1 - \frac{x}{3} + y]}{\sqrt{10}}$$

Questa possiede le soluzioni

```
| sol = Solve[equaz, y]
```

$$\{\{y \rightarrow 1 - x\}, \{y \rightarrow 1 + x\}\}$$

Dobbiamo manipolare questa espressione per trasformarla in due equazioni da assegnare ai simboli d (rappresentativo della bisettrice con coefficiente angolare negativo) e c (bisettrice di coefficiente angolare positivo) per cui

```
| {d, c} = Flatten[Apply[Equal, sol, {2}]]
```

```
{y == 1 - x, y == 1 + x}
```


Le rette c e d sono pertanto descritte dalle equazioni

```
| {c, d}
```

```
{y == 1 + x, y == 1 - x}
```

e i rispettivi coefficienti angolari evidenziano come le bisettrici siano mutuamente perpendicolari. Riportiamo tutte le rette finora trattate in un unico grafico

```
| graficoRette[{a, b, c, d}, {x, -3, 3},  
  {y, -1, 3}, PlotStyle -> colore /@ {rosso, verde, blu, magenta}];
```



Per determinare il rapporto delle aree dei triangoli avente per vertice comune il punto $R(0, 1)$ e definiti dalle terne $\{a, d, \text{asse } x\}$ e $\{b, c, \text{asse } x\}$ è sufficiente trovare quello dei segmenti individuati sull'asse x visto che le altezze dei due triangoli sono le medesime. Questi punti si ottengono

```
| {puntoA = puntoIntersezione[a, y == 0],  
  puntoB = puntoIntersezione[b, y == 0],  
  puntoC = puntoIntersezione[c, y == 0],  
  puntoD = puntoIntersezione[d, y == 0]}
```

```
{{-1/3, 0}, {-3, 0}, {-1, 0}, {1, 0}}
```

e il rapporto delle aree risulta infine

$$\frac{A(\triangle ADR)}{A(\triangle BCR)} = \frac{AD}{BC}$$


ossia

```
| distanza2Punti[puntoA, puntoD]  
  distanza2Punti[puntoB, puntoC]
```

```
2  
3
```


Una ulteriore rappresentazione grafica della situazione risulta essere la seguente

```
Show[Graphics[{{colore[oro], Polygon[{puntoB, puntoC, {0, 1}}]},
  {colore[giallo], Polygon[{puntoA, puntoD, {0, 1}}]}],
graficoRette[{a, b, c, d}, {x, -3, 3}, {y, -1, 3}, DisplayFunction->Identity,
PlotStyle->colore/@ {rosso, verde, blu, magenta}], Axes->True];
```



Come accennato sopra, notato che i coefficienti angolari delle rette a e b sono reciproci, ne segue che le medesime rette formano con il semiasse positivo delle x angoli complementari. La bisettrice c deve perciò individuare con il medesimo asse un angolo di 45° . È allora immediato determinare l'equazione di c e di conseguenza quella di d .

Esercizio 2.3

```
Clear["Global`*"]
```

▼ Dato il triangolo avente i lati sulle rette di equazione $a : y = \frac{4}{3}x$, $b : y = 4$, $c : x + y - 14 = 0$ determinare:

- le equazioni degli assi del triangolo e le coordinate del circocentro;
- i punti P del piano situati a distanza 2 dall'asse delle ordinate e $\frac{5}{2}\sqrt{10}$ dal circocentro.

a) Definiamo le tre rette del problema

$$\begin{cases} a = y = \frac{4}{3}x, \\ b = y = 4, \\ c = x + y - 14 = 0 \end{cases}$$

$$\left\{ y = \frac{4x}{3}, y = 4, -14 + x + y = 0 \right\}$$

e determiniamo i vertici del triangolo ABC come intersezioni delle rette contenenti i lati

```
{puntoA = puntoIntersezione[a, b],
 puntoB = puntoIntersezione[b, c],
 puntoC = puntoIntersezione[c, a]}
```

$$\{\{3, 4\}, \{10, 4\}, \{6, 8\}\}$$

Le equazioni degli assi verranno costruite a partire dai punti medi e dalle rette a , b e c (in alternativa si possono ottenere individuandoli come luoghi). Pertanto i punti medi dei lati sono

```
{puntoP = puntoMedio[puntoA, puntoB],
 puntoQ = puntoMedio[puntoB, puntoC],
 puntoR = puntoMedio[puntoC, puntoA]}
```

```
{{{13/2, 4}, {8, 6}, {9/2, 6}}}
```

e le equazioni dei tre assi risultano

```
{asseAB = rettaPerpendicolare[puntoP, b],
 asseBC = rettaPerpendicolare[puntoQ, c], asseCA = rettaPerpendicolare[puntoR, a]}
```

```
{-13/2 + x == 0, -2 + x - y == 0, -25/2 + x + 4/3 y == 0}
```


Per ottenere il circocentro basta intersecare due qualunque di queste rette

```
circocentro = puntoIntersezione[asseAB, asseBC]
```

```
{13/2, 9/2}
```

b) Rappresentiamo in figura la situazione geometrica appena studiata: in color rosso appaiono le rette date mentre, in verde e blu, due dei tre assi.

```
graficoRette[{a, b, c, asseAB, asseBC}, {x, 0, 15},
 {y, 0, 15}, PlotStyle -> colore /@ {rosso, rosso, rosso, verde, blu}];
```



Per determinare i punti P richiesti definiamo le loro coordinate come incognite

```
puntoP = {x0, y0}
```

```
{x0, y0}
```

e imponiamo le due condizioni proposte dal testo ossia che la loro distanza dal circocentro abbia il valore $\frac{5}{2} \sqrt{10}$

```
| eq1 = distanza2Punti[circocentro, puntoP] ==  $\frac{5}{2} \sqrt{10}$ 
```

$$\sqrt{\left(-\frac{13}{2} + x_0\right)^2 + \left(-\frac{9}{2} + y_0\right)^2} = 5 \sqrt{\frac{5}{2}}$$

così come la distanza dall'asse delle ordinate dev'essere pari a 2

```
| eq2 = Abs[x0] == 2
```

$$\text{Abs}[x_0] == 2$$

Anziché ricercare direttamente le soluzioni del sistema di queste due equazioni è dato che, per la seconda, le soluzioni appaiono immediate e cioè $x_0 = 2 \vee x_0 = -2$, determiniamo in corrispondenza i valori di y_0 . Se quindi $x_0 = 2$ la prima condizione fornisce

```
| sol1 = Solve[eq1 /. {x0 -> 2}, y0]
```

$$\{\{y_0 \rightarrow -2\}, \{y_0 \rightarrow 11\}\}$$

mentre in corrispondenza di $x_0 = -2$

```
| sol2 = Solve[eq1 /. {x0 -> -2}, y0]
```

$$\{\}$$

non otteniamo alcuna soluzione. I punti del piano richiesti sono pertanto

```
| puntoP /. sol1 /. x0 -> 2
```

$$\{\{2, -2\}, \{2, 11\}\}$$

Esercizio 2.4

```
| Clear["Global`*"]
```

- ▼ Assegnati i due punti $A(1, 2)$ e $C(2, 4)$, estremi di una diagonale di un rombo $ABCD$ di area $\frac{5}{2}$, di determinino i rimanenti due vertici e il perimetro di $ABCD$.

Definiti i vertici

```
| puntoA = {1, 2};  
| puntoC = {2, 4};
```

calcoliamo il punto medio della diagonale AC

```
| puntoM = puntoMedio[puntoA, puntoC]
```

$$\left\{ \frac{3}{2}, 3 \right\}$$

e costruiamo l'asse di AC dopo aver trovato l'equazione della retta AC

```
| rettaAC = retta2Punti[puntoA, puntoC]
```

$$-2x + y = 0$$

```
| asseAC = rettaPerpendicolare[puntoM, rettaAC]
```

$$-\frac{15}{2} + x + 2y = 0$$

Le due rette possiedono coefficienti angolari opposti e reciproci, come aspettato.

```
| coefficienteAngolare/@{rettaAC, asseAC}
```

$$\left\{ 2, -\frac{1}{2} \right\}$$

I vertici B e D del rombo appartengono a quest'ultima retta per cui, introdotte le incognite

```
| p0 = {x0, y0}
```

$$\{x0, y0\}$$

la condizione di appartenenza permette di sostituire questa coppia nell'equazione dell'asse

```
| eq1 = appartenenzaPuntoRetta[p0, asseAC]
```

$$-\frac{15}{2} + x0 + 2y0 = 0$$

La seconda condizione si ottiene imponendo che l'area abbia il valore richiesto: poiché

```
| lunghezzaAC = distanza2Punti[puntoA, puntoC]
```

$$\sqrt{5}$$

e

```
| lunghezzaPOM = distanzaPuntoRetta[p0, rettaAC]
```

$$\frac{\text{Abs}[-2x0 + y0]}{\sqrt{5}}$$

l'area del rombo risulta

```
| eq2 = lunghezzaAC lunghezzaPOM == 5 / 2
```

$$\text{Abs}[-2x_0 + y_0] == \frac{5}{2}$$

Risolvendo queste due equazioni (trascurare il messaggio del kernel di *Mathematica*)

```
| sol = Solve[{eq1, eq2}, {x0, y0}]
```

$$\left\{ \left\{ x_0 \rightarrow \frac{1}{2}, y_0 \rightarrow \frac{7}{2} \right\}, \left\{ x_0 \rightarrow \frac{5}{2}, y_0 \rightarrow \frac{5}{2} \right\} \right\}$$

si ottengono i punti *B* e *D*

```
| {puntoB, puntoD} = p0 /. sol
```

$$\left\{ \left\{ \frac{1}{2}, \frac{7}{2} \right\}, \left\{ \frac{5}{2}, \frac{5}{2} \right\} \right\}$$

Il perimetro ora si calcola facilmente

```
| perimetroABCD = 4 distanza2Punti[puntoA, puntoB]
```

$$2\sqrt{10}$$


Notiamo infine che, essendo

```
| diagonaleBD = distanza2Punti[puntoB, puntoD]
```

$$\sqrt{5}$$

le due diagonali hanno la medesima lunghezza per cui il rombo è in effetti un quadrato. Un grafico della situazione appena studiata risulta

```
Show[Graphics[{colore[giallo], Polygon /@ {{puntoA, puntoB, puntoC, puntoD}}},
Graphics[{PointSize[0.015], Point /@ {puntoA, puntoB, puntoC, puntoD}}],
graficoRette[{rettaAC, asseAC}, {x, 0, 6}, {y, 0, 6}, DisplayFunction->Identity],
DisplayFunction->$DisplayFunction, AspectRatio->Automatic, Axes->True];
```



3. Esercizi a carattere teorico

Esercizio 3.1

```
Clear["Global`*"];
```

- ▼ Dimostrare la formula dell'area di un triangolo di dati vertici utilizzando le nozioni di retta per due punti e distanza di un punto da questa.

Qui definiamo le coordinate dei tre vertici

```
pA = {xA, yA};
pB = {xB, yB};
pC = {xC, yC};
```

la retta passante per pA e pB è

```
rettaAB = retta2Punti[pA, pB]
```

$$(-xA + xB) y - xB yA + x (yA - yB) + xA yB = 0$$

mentre la base del triangolo misura

```
baseAB = distanza2Punti[pA, pB]
```

$$\sqrt{(-xA + xB)^2 + (-yA + yB)^2}$$

L'altezza rispetto alla base AB si ottiene determinando la distanza del terzo vertice C dalla retta AB

```
| altezzaCH = distanzaPuntoRetta[pC, rettaAB]
```

$$\frac{\text{Abs}[-x_B y_A + x_C (y_A - y_B) + x_A y_B + (-x_A + x_B) y_C]}{\sqrt{(-x_A + x_B)^2 + (y_A - y_B)^2}}$$

L'area del triangolo ABC è quindi

```
| areaABC = 1/2 baseAB altezzaCH // Simplify
```

$$\frac{1}{2} \text{Abs}[x_B y_A - x_C y_A - x_A y_B + x_C y_B + x_A y_C - x_B y_C]$$

Confrontiamo questo risultato con l'espressione discussa nella [parte teorica](#) e che fornisce direttamente l'area del triangolo dati i vertici

```
| areaTriangolo[pA, pB, pC] == areaABC // Simplify
```

```
True
```

Le due espressioni coincidono.

Esercizio 3.2

```
| Clear["Global`*"]
```

▼ Determinare le coordinate del baricentro note quelle dei vertici di un triangolo.

Definiti i vertici del triangolo ABC

```
| pA = {xA, yA};  
| pB = {xB, yB};  
| pC = {xC, yC};
```

determiniamo il baricentro come il punto di intersezione di due mediane. Determiniamo quindi l'equazione delle mediane AM e BN individuando innanzitutto i punti medi di BC e AC

```
| {pMedioBC = puntoMedio[pB, pC],  
| pMedioAC = puntoMedio[pA, pC]}
```

$$\left\{ \left\{ \frac{x_B + x_C}{2}, \frac{y_B + y_C}{2} \right\}, \left\{ \frac{x_A + x_C}{2}, \frac{y_A + y_C}{2} \right\} \right\}$$

Le mediane sono ora

```
| medianaAM = retta2Punti[pA, pMedioBC]
```

$$\left(-x_A + \frac{x_B + x_C}{2} \right) y - \frac{1}{2} (x_B + x_C) y_A + x \left(y_A + \frac{1}{2} (-y_B - y_C) \right) + \frac{1}{2} x_A (y_B + y_C) = 0$$

```
| medianaBN = retta2Punti[pB, pMedioAC]
```

$$\left(-x_B + \frac{x_A + x_C}{2}\right) y - \frac{1}{2} (x_A + x_C) y_B + x \left(y_B + \frac{1}{2} (-y_A - y_C)\right) + \frac{1}{2} x_B (y_A + y_C) == 0$$

Intersecandole abbiamo le coordinate del baricentro

```
| sol = Solve[{medianaAM, medianaBN}, {x, y}] // Simplify
```

$$\left\{\left\{x \rightarrow \frac{1}{3} (x_A + x_B + x_C), y \rightarrow \frac{1}{3} (y_A + y_B + y_C)\right\}\right\}$$

cioè

```
| baricentro = {x, y} /. Flatten[sol]
```

$$\left\{\frac{1}{3} (x_A + x_B + x_C), \frac{1}{3} (y_A + y_B + y_C)\right\}$$

Esercizio 3.3

```
| Clear["Global`*"];
```

- ▼ Determinare le coordinate del punto $P_1(x_1, y_1)$ simmetrico di $P(x, y)$ rispetto alla retta di equazione $r: ax + by + c = 0$.

Definiamo i punti P e P_1 e la retta r

```
| puntoP = {x, y};
| puntoP1 = {x1, y1};
| r = a x + b y + c == 0;
```

Il punto medio M del segmento PP_1

```
| puntoM = puntoMedio[puntoP, puntoP1]
```

$$\left\{\frac{x + x_1}{2}, \frac{y + y_1}{2}\right\}$$

appartiene alla retta r per cui una prima condizione risulta

```
| eq1 = appartenenzaPuntoRetta[puntoM, r]
```

$$c + \frac{1}{2} a (x + x_1) + \frac{1}{2} b (y + y_1) == 0$$

La seconda condizione si ottiene imponendo che il coefficiente angolare della retta PP_1 sia opposto e reciproco di quello di r , cosicché


```
| eq2 = coefficienteAngolare[puntoP, puntoP1] == -1 / coefficienteAngolare[r]
```

$$\frac{-y + y_1}{-x + x_1} = \frac{b}{a}$$

Risolvendo il sistema di queste due equazioni nelle incognite x_1, y_1

```
| sol = Solve[{eq1, eq2}, {x1, y1}]
```

$$\left\{ \left\{ x_1 \rightarrow -\frac{2ac + a^2x - b^2x + 2aby}{a^2 + b^2}, y_1 \rightarrow -\frac{2bc + 2abx - a^2y + b^2y}{a^2 + b^2} \right\} \right\}$$

si ottengono le coordinate del punto immagine $P_1(x_1, y_1)$ nella simmetria assiale di asse r .

```
| puntoP1 = Flatten[puntoP1 /. sol]
```

$$\left\{ \frac{-a^2x + b^2x - 2a(c + by)}{a^2 + b^2}, \frac{-2b(c + ax) + a^2y - b^2y}{a^2 + b^2} \right\}$$

Le espressioni ottenute per le coordinate di P_1 assumono forme più semplici se consideriamo le principali simmetrie assiali: tra queste abbiamo quelle con asse

- parallelo all'asse delle ascisse ossia con equazione $r: y = k$. In tal caso basta sostituire $a = 0, b = 1, c = -k$ nella precedente per ottenere

```
| puntoP1 /. {a -> 0, b -> 1, c -> -k}
```

$$\{x, 2k - y\}$$

- parallelo all'asse delle ordinate ossia con equazione $r: x = k$. In tal caso si deve sostituire $a = 1, b = 0, c = -k$

```
| puntoP1 /. {a -> 1, b -> 0, c -> -k}
```

$$\{2k - x, y\}$$

- coincidente con la bisettrice $r: y = x$ cosicché $a = 1, b = -1, c = 0$

```
| puntoP1 /. {a -> 1, b -> -1, c -> 0}
```

$$\{y, x\}$$

- oppure coincidente con la bisettrice del secondo e quarto quadrante $r: y = -x$ dove $a = 1, b = 1, c = 0$

```
| puntoP1 /. {a -> 1, b -> 1, c -> 0}
```

$$\{-y, -x\}$$

4. Esercizi sui fasci di rette

Esercizio 4.1

```
| Clear["Global`*"];
```

- ▼ Determinare le rette del fascio $f : (2 + k)x + (k - 1)y - 5 - k = 0$ che individuano con entrambi gli assi cartesiani dei triangoli di area $\frac{1}{2}$.

Studiamo le caratteristiche del fascio:

```
| f = -5 - k + (2 + k) x + (-1 + k) y == 0
```

$$-5 - k + (2 + k)x + (-1 + k)y = 0$$

escluso che sia un fascio di rette improprio in quanto il coefficiente angolare

```
| coefficienteAngolare[f]
```

$$\frac{2 + k}{1 - k}$$

dipende dal parametro, individuiamo generatrici e centro

```
| equazioneEsplicita/@generatriciFascio[f, k]
```


$$\{y = -5 + 2x, y = 1 - x\}$$

```
| pC = centroFascio[f, k]
```

$$\{2, -1\}$$

Il fascio è proprio con centro in $(2, -1)$ e generatrici $y = 2x - 5$ e $y = -x + 1$. Il grafico di alcune sue rette è

```
| graficoFascioGeneratrici[f, {x, -5, 5}, {y, -5, 5}, {k, -2, 2}];
```



Il punto di intersezione con l'asse delle x , pA , risulta

```
| pA = puntoIntersezione[f, y == 0]
```

$$\left\{ -\frac{5-k}{2+k}, 0 \right\}$$

e con l'asse delle y , pB

```
| pB = puntoIntersezione[f, x == 0]
```

$$\left\{ 0, -\frac{5-k}{-1+k} \right\}$$

L'area del triangolo formato dai punti A , B e $O(0, 0)$ è pertanto

```
| areaTriangolo[pA, pB, {0, 0}] // Simplify
```

$$\frac{1}{2} \text{Abs} \left[\frac{(5+k)^2}{-2+k+k^2} \right]$$

e posta uguale ad $\frac{1}{2}$, fornisce un'equazione nell'incognita k che si risolve

```
| equazione = areaTriangolo[pA, pB, {0, 0}] == 1/2
```

$$\frac{1}{2} \text{Abs} \left[\frac{(-5-k)^2}{(-1+k)(2+k)} \right] == \frac{1}{2}$$

```
| sol = Solve[equazione, k]
```

$$\{ \{k \rightarrow -3\} \}$$

L'unica soluzione trovata risulta $k = -3$ e la retta corrispondente è

```
| equazioneEsplicita[f /. Flatten[sol]]
```

$$y = -\frac{1}{2} - \frac{x}{4}$$

Esercizio 4.2

```
| Clear["Global`*"];
```

- ▼ Individuata la retta comune ai due fasci rappresentati dalle equazioni $f_1 : 3(k-1)x + 4(1-k)y - 8k = 0$ e $f_2 : (1+h)x + 2hy - 1 + 3h = 0$, si calcoli l'area del triangolo che tale retta delimita con gli assi cartesiani.

Dopo aver definito i due fasci

```
| f1 = 3 (k - 1) x + 4 (1 - k) y - 8 k == 0;
| f2 = (1 + h) x + 2 h y - 1 + 3 h == 0;
```

studiamone le caratteristiche scrivendo il primo nella forma esplicita

```
| equazioneEsplicita[f1]
```

$$y = -\frac{2k}{-1+k} + \frac{3x}{4}$$

Appare allora che il coefficiente angolare di f_1 è indipendente dal parametro k per cui tale fascio risulta costituito da rette parallele (mentre per $k = -1$, valore escluso dalla forma esplicita non si ottiene alcuna retta). Per il secondo, esclusa questa eventualità in quanto

```
| equazioneEsplicita[f2]
```

$$y = -\frac{-1+3h}{2h} - \frac{(1+h)x}{2h}$$

il coefficiente angolare dipende dal parametro h (ancora $h \neq 0$), determiniamo le generatrici e il centro

```
| generatriciFascio[f2, h]
```

$$\{-1+x=0, 3+x+2y=0\}$$

```
| centrof2 = centroFascio[f2, h]
```

$$\{1, -2\}$$

La retta comune richiesta non potrà che essere quella di f_2 avente il coefficiente angolare pari a quello comune di f_1 per cui dovremo risolvere l'equazione

```
| param2 = Flatten[Solve[coefficienteAngolare[f2] == coefficienteAngolare[f1], h]]
```

$$\{h \rightarrow -\frac{2}{5}\}$$

nell'incognita h . Per il primo fascio, la retta cercata dovrà passare invece per il centro del secondo. Imponiamo pertanto l'appartenenza del punto `centrof2` al primo fascio e risolviamo l'equazione nell'incognita k

```
| equaz = appartenenzaPuntoRetta[centrof2, f1]
```

$$-8 (1 - k) + 3 (-1 + k) - 8 k = 0$$

```
| param1 = Flatten[Solve[equaz, k]]
```

$$\{k \rightarrow \frac{11}{3}\}$$

In corrispondenza di questi due valori si ottiene in effetti

```
| retta1 = f1 /. param1 // Simplify
```

$$3 x = 11 + 4 y$$

e per il secondo fascio

```
| retta2 = f2 /. param2 // Simplify
```

$$3 x = 11 + 4 y$$

che, come aspettato, risulta essere la medesima retta. Per determinare l'area del triangolo che quest'ultima fa con gli assi coordinati determiniamone i punti di intersezione

```
| puntoA = puntoIntersezione[retta1, y == 0]
```

$$\{\frac{11}{3}, 0\}$$

```
| puntoB = puntoIntersezione[retta1, x == 0]
```

$$\{0, -\frac{11}{4}\}$$

per cui l'area risulta

```
| areaTriangolo[{0, 0}, puntoA, puntoB]
```

$$\frac{121}{24}$$

Esercizio 4.3

```
| Clear["Global`*"]
```

- ▼ Dopo aver individuato le caratteristiche del fascio f di rette di equazione $(1+k)x + (2k-4)y - 6k = 0$, determinare per quali valori di k le rette intersecano il segmento di estremi $P(0, 1)$ e $Q(4, 3)$.

Ad una prima analisi, il fascio

$$f = (1+k)x + (2k-4)y - 6k = 0$$

$$-6k + (1+k)x + (-4+2k)y = 0$$

appare avere il coefficiente angolare dipendente dal parametro

$$\text{coefficienteAngolare}[f]$$

$$\frac{1+k}{4-2k}$$

per cui, non essendo improprio, cerchiamone le rette sostegno e l'eventuale centro. Le prime sono

$$\{g1, g2\} = \text{generatriciFascio}[f, k]$$

$$\{x - 4y = 0, -6 + x + 2y = 0\}$$

e tra queste, la seconda $g2$ non appartiene al fascio mentre la prima si ottiene per $k = 0$: difatti la fattorizzazione del parametro implica l'equazione

$$\text{Collect}[\text{First}[f], k] = 0$$

$$x - 4y + k(-6 + x + 2y) = 0$$

che mostra come per $k \rightarrow \infty$ le rette possano solo tendere alla $g2$. Il centro di f risulta

$$\text{puntoC} = \text{centroFascio}[f, k]$$

$$\{4, 1\}$$

Definiti i punti $P(0, 1)$ e $Q(4, 3)$

$$\{\text{puntoP} = \{0, 1\}, \text{puntoQ} = \{4, 3\}\}$$

$$\{\{0, 1\}, \{4, 3\}\}$$


e determinata la retta PQ

$$\text{rettaPQ} = \text{retta2Punti}[\text{puntoP}, \text{puntoQ}]$$

$$-4 - 2x + 4y = 0$$


l'animazione del fascio

```
graficoFascioGeneratrici[f, {x, -1, 5}, {y, -1, 3}, {k, -10, 10}];
```



conferma come all'aumentare di k in valore assoluto (si provi a far variare k in intervalli di valori elevati), le rette si avvicinano a g_2 . Poiché la generatrice g_2 interseca la retta PQ , i valori del parametro k aspettati dovranno quindi assumere valori molto grandi in valore assoluto (figura successiva).

```
Show[Graphics[{Text["P", puntoP, {-1, 1}], Text["Q", puntoQ, {-1, 1}], Text["C", puntoC, {1, 1}]}],
Graphics[{PointSize[0.02], Point[puntoP], Point[puntoQ], Point[puntoC]}],
graficoRette[{g2, rettaPQ}, {x, -1, 5}, {y, -1, 5},
PlotStyle -> {RGBColor[0, 0, 1], RGBColor[0, 1, 0]}, DisplayFunction -> Identity], Axes -> True];
```



Per determinare l'insieme di variabilità di k procediamo algebricamente per cui si dovrà risolvere il sistema tra f e la retta PQ imponendo che le ascisse dei punti di intersezione appartengano all'intervallo $[0, 4]$ definito dalle ascisse di P e Q . Le coordinate del punto di intersezione si ottengono con

```
{ascissa, ordinata} = puntoIntersezione[f, rettaPQ]
```

$$\left\{ \frac{4(1+k)}{-1+2k}, -\frac{-1-4k}{-1+2k} \right\}$$

Le soluzioni si ottengono facilmente con l'istruzione

```
Reduce[0 ≤ ascissa ≤ 4, k]
```

$$k \leq -1 \mid k \geq 2$$

Come aspettato vengono coinvolti valori del parametro appartenenti ad intervalli illimitati sia con valori positivi che negativi.

Esercizio 4.4

```
| Clear["Global`*"]
```

- ▼ Nel fascio di rette di centro $C(5, 3)$ determinare la retta r perpendicolare alla retta di equazione $4x - 3y - 1 = 0$ e la retta s che, con l'origine, intercetta sul semiasse x positivo un segmento di lunghezza 2. Calcolare infine l'equazione della retta t parallela all'asse y che delimita con le rette r e s un triangolo di area $\frac{63}{2}$.

Definito il punto $C(5, 3)$

```
| puntoC = {5, 3}
```

```
{5, 3}
```

costruiamo il fascio proprio di centro C

```
| f = y - 3 == m (x - 5)
```

```
-3 + y == m (-5 + x)
```

Poiché la retta r

```
| rettaData = 4 x - 3 y - 1 == 0
```

```
-1 + 4 x - 3 y == 0
```

possiede coefficiente angolare

```
| mData = coefficienteAngolare[rettaData]
```

```
 $\frac{4}{3}$ 
```

la retta r si ottiene dal fascio f sostituendovi l'opposto del reciproco

```
| r = f /. (m -> -1 / mData)
```

```
 $-3 + y == -\frac{3}{4} (-5 + x)$ 
```

In forma esplicita risulta

```
| r = equazioneEsplicita[r]
```

```
 $y == \frac{27}{4} - \frac{3 x}{4}$ 
```

La retta s invece, si deduce imponendo il passaggio di f per il punto $(2, 0)$ in quanto deve intercettare sul semiasse positivo delle ascisse un segmento (con secondo estremo l'origine) di lunghezza 2


```
| condizione = appartenenzaPuntoRetta[{2, 0}, f]
```

$$-3 == -3 m$$

da cui

```
| ms = Flatten[Solve[condizione, m]]
```

$$\{m \rightarrow 1\}$$

Sostituendo tale valore in f si ha

```
| s = equazioneEsplicita[f /. ms]
```

$$y == -2 + x$$

Dobbiamo ora definire il fascio di rette parallele all'asse y

```
| t = x == k
```

$$x == k$$


e i suoi punti di intersezione con r e s

```
| {puntoA = puntoIntersezione[t, r],  
   puntoB = puntoIntersezione[t, s]}
```

$$\left\{ \left\{ k, -\frac{3}{4}(-9 + k) \right\}, \{k, -2 + k\} \right\}$$

L'animazione seguente fornisce una rappresentazione della situazione geometrica

```
| graficoAnimazioneFascio[{t, r, s}, {x, -1, 7}, {y, -3, 5}, {k, -1, 6}];
```



L'area di $\triangle ABC$ può essere determinata per mezzo della formula che la esprime in termini dei tre vertici oppure, come faremo in questo caso, individuando le lunghezze della base AB e della relativa altezza CH .

```
| baseAB = distanza2Punti[puntoA, puntoB] // Simplify
```

$$\frac{7}{4} \sqrt{(-5+k)^2}$$

L'altezza relativa alla base AB è invece la differenza in valore assoluto delle ascisse di C e di un qualsiasi punto di t :

```
| altezzaCH = Abs[5 - k]
```

$$\text{Abs}[5 - k]$$

di conseguenza risulta

```
| areaABC = 1/2 baseAB altezzaCH
```

$$\frac{7}{8} \sqrt{(-5+k)^2} \text{Abs}[5 - k]$$

per cui, posta uguale al valore dato dal testo, risolviamo la conseguente equazione

```
| sol = Solve[areaABC == 63/2, k]
```

$$\{\{k \rightarrow -1\}, \{k \rightarrow 11\}\}$$

In definitiva, le due equazioni corrispondenti per le rette t sono

```
| x == k /. sol
```

$$\{x == -1, x == 11\}$$

Esercizio 4.5

```
| Clear["Global`*"]
```

▼ Studiate le caratteristiche dei due fasci

$$a : 5 - \frac{7k}{3} + \left(-1 + \frac{k}{3}\right)x + (-2 + k)y == 0 \text{ e}$$

$$b : 8 - \frac{10k}{3} + \left(-4 - \frac{k}{3}\right)x + (4 - k)y == 0,$$

determinare il luogo Γ dei loro punti di intersezione.

Le caratteristiche si studiano facilmente osservando dapprima i coefficienti angolari e, nel caso di dipendenza dal parametro, ricercando l'eventuale centro del fascio e le corrispondenti generatrici. Per determinare invece il luogo richiesto otterremo innanzitutto le coordinate del loro punto di intersezione in funzione del parametro k per poi interpretarle come una coppia di equazioni parametriche rappresentative del luogo Γ .

Definiamo quindi i due fasci

$$\left\{ \begin{array}{l} a = 5 - \frac{7k}{3} + \left(-1 + \frac{k}{3}\right)x + (-2+k)y = 0, \\ b = 8 - \frac{10k}{3} + \left(-4 - \frac{k}{3}\right)x + (4-k)y = 0 \end{array} \right\}$$

$$\left\{ 5 - \frac{7k}{3} + \left(-1 + \frac{k}{3}\right)x + (-2+k)y = 0, 8 - \frac{10k}{3} + \left(-4 - \frac{k}{3}\right)x + (4-k)y = 0 \right\}$$

Le caratteristiche discendono facilmente osservando innanzitutto i rispettivi coefficienti angolari

$$\{ \text{coefficienteAngolare}[a], \text{coefficienteAngolare}[b] \}$$

$$\left\{ \frac{3-k}{-6+3k}, \frac{12+k}{12-3k} \right\}$$

e, poiché questi presentano una dipendenza (non eliminabile) dal parametro, possiamo escludere che i fasci proposti siano costituiti da rette parallele cioè impropri. Vediamo se possiedono un centro e delle generatrici

$$\{ \text{centro1} = \text{centroFascio}[a, k], \\ \text{gener1} = \text{generatriciFascio}[a, k] \}$$

$$\{ \{1, 2\}, \{5 - x - 2y = 0, -\frac{7}{3} + \frac{x}{3} + y = 0\} \}$$

$$\{ \text{centro2} = \text{centroFascio}[b, k], \\ \text{gener2} = \text{generatriciFascio}[b, k] \}$$

$$\{ \{-1, -3\}, \{8 - 4x + 4y = 0, -\frac{10}{3} - \frac{x}{3} - y = 0\} \}$$

Sono pertanto fasci di rette propri. Il loro punto di intersezione P si ottiene

$$\text{puntoP} = \text{puntoIntersezione}[a, b]$$

$$\left\{ 3 - k, -\frac{1}{3}(-3 - k) \right\}$$


Al variare del parametro nell'insieme dei numeri reali, ciascuna delle coordinate di tale punto mostra una dipendenza lineare da k . Abbiamo cioè una rappresentazione parametrica dell'insieme Γ richiesto e, dato che appare descritto da relazioni lineari, Γ non potrà che essere una retta. Difatti esplicitato il legame tra ascissa e ordinata dei punti appartenenti a tale luogo

$$\text{eqParam} = \text{Thread}[\text{Equal}[\{x, y\}, \text{puntoP}]]$$

$$\{x = 3 - k, y = -\frac{1}{3}(-3 - k)\}$$

abbiamo nient'altro che la tipica rappresentazione parametrica di una retta: il grafico di questo luogo assieme ai due fasci mostra con evidenza questi risultati

```
| graficoRetteParametriche[{eqParam}, k, {x, -4, 4}, {y, -1, 4}];
```



Volendo riportarci alla forma implicita, cioè ad un'equazione che metta in relazione direttamente l'ascissa e l'ordinata dei punti di Γ , va eliminata la dipendenza dal parametro dalla coppia di equazioni ottenute sopra. Ciò si ottiene esplicitando il parametro da una equazione

```
| param = Flatten[Solve[eqParam[[1]], k]]
```

```
{k -> 3 - x}
```

e quindi sostituendolo nell'altra

```
| luogoGamma = eqParam[[2]] /. param
```

$$y == \frac{6 - x}{3}$$

Passando alla rappresentazione esplicita abbiamo

```
| luogoGamma = equazioneEsplicita[luogoGamma]
```

$$y == 2 - \frac{x}{3}$$


Tutto ciò può eseguirsi anche in un unico passaggio con l'istruzione

```
| luogoGamma = equazioneEsplicita[riduceParametrica[eqParam, k]]
```

$$y == 2 - \frac{x}{3}$$

Proponiamo infine un'animazione grafica che riassume visivamente quanto trovato: appare allora evidente come, al variare del parametro k , il punto di intersezione dei due fasci appartenga alla retta (in rosso) trovata.

```
graficoAnimazioneFascio[{a, b, luogoGamma}, {x, -5, 5}, {y, -5, 5}, {k, -2, 8, .5},
  PlotStyle -> {colore[blu], colore[blu], {colore[rosso], Thickness[0.007]}}];
```



Esercizio 4.6

```
Clear["Global`*"];
```

- ▼ Sono assegnati i fasci $\alpha : (y + 3)(k + 2) = (3k + 1)(x + 1)$, $\beta : (y + 3)(3k + 1) = -(k + 2)(x + 1)$. Studiate le caratteristiche di ciascuno e i legami reciproci, siano A e B le intersezioni, rispettivamente, di α e β con l'asse delle ascisse. Determinare k in modo che valga la relazione $\frac{1}{15} A(ACB) = 3$, con C centro comune di α e β .

Definiamo i due fasci

```
 $\alpha = (y + 3)(k + 2) == (3k + 1)(x + 1);$ 
 $\beta = (y + 3)(3k + 1) == -(k + 2)(x + 1);$ 
```

e studiamone le caratteristiche

```
centroFascio[ $\alpha$ , k]
```

```
{-1, -3}
```

```
centroC = centroFascio[ $\beta$ , k]
```

```
{-1, -3}
```

Come detto dal testo, hanno pertanto il medesimo centro $C(-1, -3)$. Inoltre esplicitando i coefficienti angolari di ciascuno

```
| {coefficienteAngolare[α],  
  coefficienteAngolare[β]}
```

$$\left\{ \frac{1+3k}{2+k}, \frac{2+k}{-1-3k} \right\}$$


si può osservare che sono tali che, per $k \neq -2$ e $k \neq -\frac{1}{3}$, il loro prodotto risulta essere

```
| coefficienteAngolare[α] coefficienteAngolare[β] // Simplify
```

```
-1
```

per cui ad ogni retta di un fascio corrisponde nell'altro una retta perpendicolare. L'animazione seguente mostra tali caratteristiche.

```
| graficoAnimazioneFascio[{α, β}, {x, -5, 5}, {y, -5, 5}, {k, -3, 3, 0.5}, centroC];
```



Determiniamo le coordinate dei punti pA e pB , rispettivamente intersezioni di α e β con l'asse delle ascisse.

```
| pA = puntoIntersezione[α, y == 0]
```

$$\left\{ \frac{5}{1+3k}, 0 \right\}$$

```
| pB = puntoIntersezione[β, y == 0]
```

$$\left\{ -\frac{5(1+2k)}{2+k}, 0 \right\}$$

Il calcolo dell'area mostra ora come questa sia una funzione del parametro k

```
| areaABC = areaTriangolo[pA, pB, centroC] // Simplify
```

$$\frac{45}{2} \text{ Abs} \left[\frac{1+2k+2k^2}{2+7k+3k^2} \right]$$

Imponendo la condizione data dal problema

$$\text{equazione} = \frac{1}{15} \text{areaABC} == 3$$

$$\frac{3}{2} \text{Abs}\left[\frac{1+2k+2k^2}{2+7k+3k^2}\right] == 3$$

e risolvendo in k otteniamo

```
valoriParam = Solve[equazione, k]
```

$$\left\{\left\{k \rightarrow \frac{1}{4}(-4 - \sqrt{6})\right\}, \left\{k \rightarrow \frac{1}{2}(-3 - \sqrt{6})\right\}, \left\{k \rightarrow \frac{1}{4}(-4 + \sqrt{6})\right\}, \left\{k \rightarrow \frac{1}{2}(-3 + \sqrt{6})\right\}\right\}$$


quattro valori tutti accettabili. Pur non richiesto dal problema, in corrispondenza di questi si hanno le rette

```
rette = Flatten[{α, β} /. valoriParam // Simplify]
```

$$\left\{20 + (8 + 3\sqrt{6})x = (-4 + \sqrt{6})y, -\frac{1}{4}(8 + 3\sqrt{6})(3 + y) = \frac{1}{4}(-4 + \sqrt{6})(1 + x),\right. \\ 10 + 7x + 3\sqrt{6}x + y = \sqrt{6}y, 10(2 + \sqrt{6}) + \sqrt{6}x + (7 + 3\sqrt{6})y = x, \\ (-8 + 3\sqrt{6})x = 20 + (4 + \sqrt{6})y, \frac{1}{4}(-8 + 3\sqrt{6})(3 + y) = -\frac{1}{4}(4 + \sqrt{6})(1 + x), \\ \left.3\sqrt{6}x = 10 + 7x + y + \sqrt{6}y, (1 + \sqrt{6})x + \sqrt{6}(10 + 3y) = 20 + 7y\right\}$$

il cui grafico è il seguente.

```
graficoRette[rette, {x, -40, 40}, {y, -40, 40}];
```



Questo mostra solo 4 rette. In effetti, va notato che i quattro valori del parametro danno origine a otto rette ma di queste solo quattro sono distinte. Le rimanenti possiedono i coefficienti angolari scambiati: il calcolo di questi ultimi rende evidente questo fatto.

```
{coefficienteAngolare[α], coefficienteAngolare[β]} /. valoriParam // N
```

$$\{-9.89898, 0.101021\}, \{9.89898, -0.101021\}, \{-0.101021, 9.89898\}, \{0.101021, -9.89898\}$$

Si osserva lo stesso fatto mappando la funzione `equazioneEsplicita` e riportando la lista delle rette in forma numerica e tabellare

```
| equazioneEsplicita /@ rette // N // TableForm
```

```
y = -12.899 - 9.89898 x
y = -2.89898 + 0.101021 x
y = 6.89898 + 9.89898 x
y = -3.10102 - 0.101021 x
y = -3.10102 - 0.101021 x
y = 6.89898 + 9.89898 x
y = -2.89898 + 0.101021 x
y = -12.899 - 9.89898 x
```

Esercizio 4.7

```
| Clear["Global`*"]
```

▼ Nel piano xOy sono dati i punti $A(5, 1)$ e $B(1, -3)$ e la retta $r: y = 3x + 10$.

- Determinare su r il punto C equidistante da A e B ;
- determinare l'equazione della retta t parallela alla retta AB che intersechi i lati AC e BC del triangolo ABC rispettivamente in D e E e abbia da AB distanza uguale ai $\frac{2}{5}$ dell'altezza CM del triangolo;
- determinare il centro della circonferenza circoscritta al trapezio $ABDE$.

a) Definiti i punti A e B e la retta r

```
| {puntoA = {5, 1}, puntoB = {1, -3},
  r = y == 3 x + 10}
```

```
{ {5, 1}, {1, -3}, y == 10 + 3 x }
```

il punto incognito $C(x_0, y_0)$ deve appartenere ad r e all'asse del segmento AB . Troviamo quindi questa retta individuando prima il punto medio M di AB

```
| puntoM = puntoMedio[puntoA, puntoB]
```

```
{3, -1}
```

e quindi la retta AB

```
| rettaAB = retta2Punti[puntoA, puntoB]
```

```
-16 + 4 x - 4 y == 0
```

L'asse si ottiene come

```
| rettaCM = rettaPerpendicolare[puntoM, rettaAB] // Simplify
```

```
x + y == 2
```


Intersecando la retta data r e la retta CM e risolvendo il sistema delle due equazioni otteniamo il punto cercato

```
| puntoC = puntoIntersezione[r, rettaCM]
```

```
{-2, 4}
```

La retta CM si sarebbe potuta determinare anche senza trovare l'equazione della retta AB ma solo con il coefficiente angolare di quest'ultima: i passaggi alternativi sono i seguenti

```
| mAB = coefficienteAngolare[puntoA, puntoB]
```

```
1
```

```
| rettaCM = rettaPerpendicolare[puntoM, mAB]
```

```
-2 + x + y == 0
```

b) Riportando la retta AB nella forma esplicita

```
| rettaAB = equazioneEsplicita[rettaAB]
```

```
y == -4 + x
```

costruiamo il fascio f di rette parallele a questa

```
| f = y == x + k
```

```
y == k + x
```

Calcolata la distanza di C dalla retta AB

```
| distCM = distanzaPuntoRetta[puntoC, rettaAB]
```

```
5  $\sqrt{2}$ 
```

imponiamo che le rette di f abbiano distanza da A pari ai $\frac{2}{5} CM$

```
| eq1 =  $\frac{2}{5}$  distCM == distanzaPuntoRetta[puntoA, f]
```

```
2  $\sqrt{2}$  ==  $\frac{\text{Abs}[-4 - k]}{\sqrt{2}}$ 
```

e quindi risolviamo l'equazione nell'incognita k

```
| Solve[eq1, k]
```

```
{{k -> -8}, {k -> 0}}
```


Una figura rappresentativa della situazione geometrica aiuta a comprendere come, tra questi valori, solo il secondo sia accettabile in quanto l'altro fornisce una retta che non interseca i lati del triangolo ABC . Difatti quest'ultima

appartiene al semipiano inferiore nel quale AB suddivide il piano cartesiano: nel grafico, assieme alle rette AC e BC tracciamo pure la retta $y = x$ appena trovata.

```
{rettaDE = y == x,
  rettaAC = retta2Punti[puntoC, puntoA], rettaBC = retta2Punti[puntoB, puntoC]}
```

```
{y == x, -22 + 3 x + 7 y == 0, -2 - 7 x - 3 y == 0}
```

```
graficoRette[{rettaAC, rettaBC, rettaAB, rettaDE}, {x, -8, 10}, {y, -8, 6},
  PlotStyle -> colore /@ {rosso, verde, marrone, blu}, AspectRatio -> Automatic];
```



c) Il centro della circonferenza circoscritta al trapezio $ABDE$ appartiene all'asse del segmento AB cioè alla retta CM (si veda la [figura](#) sottostante) e contemporaneamente all'asse del segmento AD . Per trovare quest'ultimo procediamo calcolando le coordinate del punto D , intersezione della retta per A e C con la retta DE

```
puntoD = puntoIntersezione[rettaAC, rettaDE]
```

```
{ 11/5, 11/5 }
```

Il punto medio del segmento AD risulta

```
puntoMedioAD = puntoMedio[puntoA, puntoD]
```

```
{ 18/5, 8/5 }
```

e l'equazione dell'asse invece

```
asseAD = rettaPerpendicolare[puntoMedioAD, rettaAC]
```

```
- 102/5 + 7 x - 3 y == 0
```

Le coordinate del centro si possono infine calcolare come le soluzioni del sistema $CM \cap asseAD$

```
centroCC = puntoIntersezione[rettaCM, asseAD]
```

```
{ 66/25, -16/25 }
```

Seguono le istruzioni per ottenere la rappresentazione grafica del problema discusso.

```
Show[Graphics[{{PointSize[0.02], Point /@ {puntoMedioAD, centroCC}},
  Text["A", puntoA, {0, 1}], Text["B", puntoB, {1, 0}], Text["D", puntoD, {0, -1}],
  Text["C", puntoC, {-1, -1}], Text["M", puntoM, {0, 1}}],
graficoRette[{rettaAC, rettaBC, rettaAB, rettaDE, rettaCM, asseAD}, {x, -4, 8}, {y, -6, 6},
  PlotStyle -> colore /@ {rosso, verde, marrone, blu, oro, violetto}, DisplayFunction -> Identity],
DisplayFunction -> $DisplayFunction, Axes -> True, AspectRatio -> Automatic];
```

